APL Version of Peter Markstein's McKeeman Table Routine

NOTES: INPUT IS A VECTOR, G, WHICH REPRESENTS A GRAMMAR, ENCODED AS A CHARACTER STRING. A PRODUCTION, A←BCD WOULD BE ENCODED ABCD| FOR EXAMPLE. THE FIRST LINE COMPUTES C, THE VECTOR OF UNIQUE CHARACTERS OCCURRING IN THE GRAMMAR. S BECOMES THE VECTOR OF LHS OF PRODUCTIONS, B THE FIRST ELEMENTS ON THE RHS OF PRODUCTIONS, E THE LAST ELEMENT OF THE PRODUCTIONS. BEGINS[I;J] IS SET TO 1 IF THE JTH SYNTACTIC TYPE CAN BEGIN WITH THE ITH SYNTACTIC TYPE; LIKEWISE FOR ENDS[I;J]. THE MATRIX T, ALSO EXTERNAL TO PRECEDENCE, HOLDS THE RESULT:

$$T[I;J] = \begin{cases} 1 & \text{IF } C[I] = C[J] \\ 2 & \text{IF } C[I] > C[J] \\ 3 & \text{IF } C[I] < C[J] \\ 5 & \text{IF } C[I],C[J] \text{ IS AMBIGUOUS} \\ 0 & \text{IF } C[I],C[J] \text{ IS UNGRAMMATICAL.} \end{cases}$$

```
∇ PRECEDENCE
      X←ρC←(⍱/(Q∘.<Q←⍳ρZ)∧Z∘.=Z)/Z←(G≠'|')/G
      T←BEGINS←ENDS←(X,X)ρ0
      S←C⍳G[1+Q←0,K[⍳¯1+ρK←(G='|')/⍳ρG]]
      B←C⍳G[2+Q]
      E←C⍳G[K-1]
      BEGINS[B;S]←I←J←1
      ENDS[E;S]←1
      BEGINS←COMPLETE BEGINS
      ENDS←COMPLETE ENDS
     XM: T[I;J]←C[I] NEXTTO C[J]
      Z←(C[I] SMALL C[J])+C[I] BIGGER C[J]
      T[I;J]←((~K)×T[I;J]+Z)+5×K←0≠T[I;J]×Z
      →((ρC)≥J←J+1)/XM
      →((ρC)≥I←I+J←1)/XM∇
∇Z←COMPLETE X;Y;Q
      Q←+/+/Z←X
     LCO: Y←Q
      Z←Z∨.∧Z
      →(Y≠Q←+/+/Z)/LCO∇
∇Z←A NEXTTO B
      Z←∨/(0,0,(G≠'|'))∧(0,(G=A),0)∧(G=B),0,0∇
∇Z←A BIGGER B;U
      U←(BEGINS[C⍳B;]/C),(~BEGINS[B;B])ρB
      Z←2×∨/(0,0,(G≠'|'))∧(0,(∨/G∘.=ENDS[C⍳A;]/C),0)∧((∨/G∘.=U),0,0)∇
∇Z←A SMALL B
      Z←3×∨/(0,0,(G≠'|'))∧(0,(G=A),0)∧((∨/G∘.=BEGINS[C⍳B;]/C),0,0)∇
```